An implementation of the Ho-Lee model for pricing short term interest rate options

Author: Filippo Ippolito

Abstract

In this work we implement a modified version of the Ho and Lee model for the valuation of interest rate options. The main difference from the original model lies in the fact that, while Ho and Lee are interested in valuing a broad range of contingent claims, we aim at pricing a specific class of options, namely options written on short term interest rate futures. This allows us to estimate the parameters of the model in a significantly simpler way and to obtain the pricing functions for European and American options.

Introduction

In recent years a number of studies has focused on a new approach to the valuation of term structure movements.

In a seminal paper Ho and Lee [Ho and Lee 1986] have developed a methodology that takes the complete term structure as given and that, imposing arbitrage-free conditions and using martingale probabilities, derives the stochastic process followed by zero coupon bonds. In this way the term structure movements, that they obtain, are a function of the initial curve and are independent from risk preferences of market agents.

These peculiar features distinguish this approach from those previously adopted by Cox, Ingersoll and Ross [1985a,1985b], Brennan and Schwartz [1977,1979,1982], Vasicek [1977], Dothan [1978], Richard [1978], Langetieg [1980], Rendleman and Bartter [1980], Courtadon [1982], Ball and Torous [1983], Artzner and Delbaen [1989], Longstaff [1989].

Ho and Lee's approach to contingent claims valuation has been adopted by an increasing number of authors, such as Bliss and Ronn [1989], Heath, Jarrow and Morton [1990,1991,1992], Hull and White [1990], Turnbull and Milne [1991], Shirakawa [1991].

In this work we implement a modified version of the Ho and Lee model. The main difference from the original model lies in the fact that, while Ho and Lee are interested in valuing a broad range of contingent claims, we aim at pricing specific class options, namely options written on short term interest rate futures. Examples of this kind are the options on futures contracts exchanged at LIFFE (London International Financial Futures and Options Exchange) on the Euro [1]. In these options the bond underlying the future is a three-month zero coupon and the expiration date of the future is the same of the option.

Ho-Lee suggest that one should calculate the implicit parameter of their model from the market yield curve [2] with a non-linear estimation process. In our case, as we are interested in a specific class of contingent claims, the estimation procedure has been significantly simplified.

In the first section we outline the Ho-Lee framework and we define the functions for pricing European and American options.The second and third sections are devoted to the implementation of the model respectively for European and American options.

The two notebooks (*Ho-Lee model_European Options.nb* and *Ho-Lee model_American Options.nb*) contain the complete list of the implemented functions with brief operational descriptions.

[1] See LIFFE Publications [1995], [1998].

[2] See Fisher and Zervos [1996] for an analysis of the yield curve with Mathematica.

The Ho-Lee framework

If **P(t,τ; i)** is the price of a zero coupon bond in the state i at time t, valued at time 0, that matures after τ periods, the evolution at time 1 of P(0, τ; i), i.e. the bond at time 0, is as follows:

$$P(0, \tau; i) = \begin{cases} P(1, \tau; i+1) \rightarrow \text{in case of an upward movement} \\ P(1, \tau; i) \rightarrow \text{in case of an downward movement} \end{cases} \qquad (1)$$

By imposing arbitrage-free conditions, risk neutrality and path-independency, Ho-Lee demonstrate that after t steps the bond price $\mathbf{P(t,\ \tau;\ i)}$ can be described by the equation:

$$P(t,\ \tau;\ i) = \frac{P(0,\ \tau+t)}{P(0,\ t)} \times (h^*(\tau+t-1)\, h^*(\tau+t-2)\ldots h^*(\tau+i)\, h(\tau+i-1)\ldots h(\tau)) / (h^*(t-1)\, h^*(t-2)\ldots h^*(i)\, h(i-1)\ldots h(1)) \tag{2}$$

In equation (2), h(τ) and h'(τ) are the *perturbation functions* of a bond with maturity τ and are defined in the following way:

$$h(\tau) = \frac{1}{\pi + (1-\pi)\,\kappa^{\tau}} \tag{3}$$

and

$$h'(\tau) = \frac{k^{\tau}}{\pi + (1-\pi)\,\kappa^{\tau}} \tag{4}$$

In the above equations, π is the *implied binomial probability*, also known as the martingale probability, that operates in a risk neutrality context, and κ determines the spread between the two perturbation functions, h(τ) and h'(τ).

Equation (2) can be simplified so to give:

$$P(t,\ \tau;\ i) = \frac{P(0,\tau+t)}{P(0,t)} \times \frac{h(\tau+t-1)\, h(\tau+t-2)\ldots h(\tau)\,\kappa^{\tau(t-i)}}{h(t-1)\, h(t-2)\ldots h(1)} \tag{5}$$

Equation (5) can be further simplified, because, in order to avoid arbitrage opportunities, the first factor must be equal to F(t, τ), the price at time t of a forward zero coupon bond with maturity τ periods, valued at time zero [3].

$$P(t,\ \tau;\ i) = F(t,\ \tau) \times \frac{h(\tau+t-1)\, h(\tau+t-2)\ldots h(\tau)\,\kappa^{\tau(t-i)}}{h(t-1)\, h(t-2)\ldots h(1)} \tag{6}$$

Equation (6) is very important to understand how the stochastic process of bond prices works. If we remember that the valuation is made at time zero, the price of a zero coupon bond at time t, where t > 0, with maturity τ is given by the price of a future, that matures at time t, on a bond with maturity τ, multiplied by all the perturbation functions relative to the periods that go from 1 to (t + τ -1) [4].

Given that:

$$h(\tau+t-1)\, h(\tau+t-2)\ldots h(\tau) = \prod_{j=1}^{t} \left(\frac{1}{\pi + (1-\pi)\,\kappa^{\tau+t-j}}\right) \tag{7}$$

and

$$h(t-1)\, h(t-2)\ldots h(1) = \prod_{m=1}^{t-1} \left(\frac{1}{\pi + (1-\pi)\,\kappa^{t-m}}\right) \tag{8}$$

we can write equation (6) as:

$$P(t, \tau; i) = F(t, \tau) \times \frac{\prod_{j=1}^{t} \left(\frac{1}{\pi + (1-\pi)\ \kappa^{\tau+t-j}}\right) \kappa^{\tau\ (t-i)}}{\prod_{m=1}^{t-1} \left(\frac{1}{\pi + (1-\pi)\ \kappa^{t-m}}\right)} \qquad (9)$$

The model is completely specified for given constants κ and π. If for simplicity we assume that the martingale probabilitieπ and (1-π) are equal to 0.5, equation (9) can be simplified to:

$$P(t, \tau; i) = F(t, \tau) \times \frac{\prod_{m=1}^{t-1} (0.5 + 0.5\,\kappa^{t-m})\ \kappa^{\tau\ (t-i)}}{\prod_{j=1}^{t} (0.5 + 0.5\,\kappa^{\tau+t-j})} \qquad (10)$$

From equation (1) we have that at time t, there will be t+1 prices, binomially distributed with probability 0.5^t. At time t we apply the option valuation function to each price, so that we will have a series of values that are either positive or zero. Their sum is then discounted at time 0 at the continuous rate r. The option valuation functions are given by the following equations for call and put options respectively, where S is the strike price of the option:

$$C = \sum_{i=0}^{t} \left[\mathrm{Max}\left\{F(t, \tau) \times \frac{\prod_{m=1}^{t-1} (0.5 + 0.5\,\kappa\text{^}(t-m))\ \kappa\text{^}(\tau\ (t-i))}{\prod_{j=1}^{t} (0.5 + 0.5\,\kappa\text{^}(\tau + t - j))} - S;\ 0\right\}\quad 0.5\text{^}t\quad e\text{^}(-r\,t/360)\quad \mathrm{Binomial}(t, i)\right] \qquad (11)$$

and

$$P = \sum_{i=0}^{t} \left[\mathrm{Max}\left\{S - F(t, \tau) \times \frac{\prod_{m=1}^{t-1} (0.5 + 0.5\,\kappa\text{^}(t-m))\ \kappa\text{^}(\tau\ (t-i))}{\prod_{j=1}^{t} (0.5 + 0.5\,\kappa\text{^}(\tau + t - j))};\ 0\right\}\quad 0.5\text{^}t\quad e\text{^}(-r\,t/360)\quad \mathrm{Binomial}(t, i)\right] \qquad (12)$$

[3] Therefore F(t,τ) matures t + τ periods after 0.

[4] See equation (2).

Implementation of the model for European Options

Let's define the following general price function [5]:

```
HoLeeEuropean[F_, t1_, τ1_, κ_, r_, exercise_Function] :=
 With[{τ = Round[τ1], t = Round[t1]},
 Sum[exercise[F*Product[0.5 + 0.5*κ^(t - m), {m, 1, t - 1}] /
   Product[0.5 + 0.5*κ^(t + τ - j), {j, 1, t}] *
   κ^(τ*(t - i))]*0.5^t*Exp[-r*t/360]*Binomial[t, i],
  {i, 0, t}]]
```

We can then have the option valuation functions for call and put options:

```
HoLeeEuropeanCall[F_, S_, t_, τ_, κ_, r_] :=
 HoLeeEuropean[F, t, τ, κ, r, Max[#1 - S, 0] &]

HoLeeEuropeanPut[F_, S_, t_, τ_, κ_, r_] :=
 HoLeeEuropean[F, t, τ, κ, r, Max[S - #1, 0] &]
```

Figure 1 shows how the price of an European call option varies, in relation to changes in the price of the underlying future and in time.

Figure .1 : The price of an European call option relatively to changes in the price of the underlying future and in time.



We then define the functions that are needed to determine the implicit parameter κ. The inputs of these functions are to be taken from quoted options, similar to the option that we want to price.

```
ImpliedKappaHoLeeEuropeanCall[
  F_, S_, t_, τ_, r_, optionprice_] :=
κ /. FindRoot[ HoLeeEuropeanCall[F, S, t, τ, κ, r] ==
optionprice, {κ, 0.99, 0.9999}]
```

```
ImpliedKappaHoLeeEuropeanPut[F_, S_, t_, τ_, r_, optionprice_] :=
κ /. FindRoot[HoLeeEuropeanPut[F, S, t, τ, κ, r] ==
optionprice, {κ, 0.99, 0.9999}]
```

[5] See Miller[1993] and Benninga, Steinmetz and Stroughair[1996] for option valuation with Mathematica.

Implementation of the model for American Options

The distinguish feature of American options is that they can be exercised at any time before expiration. Therefore we implement a recursive function, that at each node of the binomial tree compares the option value, at that time and in that state, with the expected value of the branches that originate from such node. This function is the following:

```
HoLeeAmerican[F_, t1_, τ1_, κ_, r_, exercise_Function] :=
Module[{OpRecurse, res, t, τ},
  t = Round[t1]; τ = Round[τ1];
OpRecurse[node_, level_] :=
   OpRecurse[node, level] = If[level == t,
            exercise[F  Product[  0.5 + 0.5 κ^( t - m) , {m, 1, t -
            1 } ] / Product[ 0.5 + 0.5 κ^ ( t  + τ - j ) , {j, 1, t}  ]
κ^(  τ (t -  node )  ) ] 0.5^t, Max[0.5  Exp[ - r / 360 ]
       ( OpRecurse[ node,  level  +  1 ] +
         OpRecurse[ node  + 1,  level  + 1] ) , exercise[
        F Product[ 0.5  + 0.5 κ^(level - m), {m, 1, level - 1 } ] /
          Product[ 0.5 + 0.5 κ^(level + τ - j ), {j, 1, level} ]
         κ^ (τ (level - node)) ] 0.5^level ] ];
  res = OpRecurse[ 0, 0];  Clear[ OpRecurse];
  res ];
```

From the general function we define the two pricing functions for call and put options:

```
HoLeeAmericanCall[F_, S_, t_, τ_, κ_, r_] :=
 HoLeeAmerican[F, t, τ, κ, r, Max[# - S,  0]  & ]
```

```
HoLeeAmericanPut[F_, S_, t_, τ_, κ_, r_] :=
 HoLeeAmerican[F, t, τ, κ, r, Max[S - #,  0]  & ]
```

Finally, as for European options, we define the functions for the determination of κ, required in the previous functions:

```
ImpliedKappaHoLeeAmericanCall[F_, S_, t_, τ_, r_, optionprice_] :=
 κ /. FindRoot[HoLeeAmericanCall[F, S, t, τ, κ, r] == optionprice,
   {κ, 0.99, 0.9999}]

ImpliedKappaHoLeeAmericanPut[F_, S_, t_, τ_, r_, optionprice_] :=
 κ /. FindRoot[HoLeeAmericanPut[F, S, t, τ, κ, r] == optionprice,
   {κ, 0.99, 0.9999}]
```

Figure 3 shows how the price of an American call option varies, accordingly to changes in the price of the underlying future and in time.

Figure .3 : Price variations of an American call option, changing the price of the underlying future and time.



Bibliography

▪ Artzner, P. e Delbaen, F. (1989). "Term Structure of Interest Rates: the Martingale Approach." *Advanced Applied Mathematics* 10, 95-129

▪ Ball, C. e Torous, W. (1983). "Bond Pricing Dynamics and Options." *Journal of Financial and Quantitative Analysis* 18, 517-531

▪ Benninga, S., Steinmetz, R. e Stroughair J. "Implementing Numerical Option Pricing Models." Sta in Hal R. Varian. " Computational Economics and Finance. Modeling and Analysis with Mathematica." TELOS, The Electronic Library of Science, Santa Clara, CA, 1996

▪ Bliss, Jr., R. e Ronn E. (1989). "Arbitrage-based Estimation of Non-stationary Shifts in the Term Structure of Interest Rates." *The Journal of Finance* 44, 591-610

▪ Brennan, M. J., and E. S. Schwartz (March 1977). "Saving Bonds, Retractable Bonds and Callable Bonds." *Journal of Financial Economics* 5, 67-88

▪ Brennan, M. J., and E. S. Schwartz (1979). "A Continous-Time Approach to the Pricing of Bonds." *Journal of Banking and Finance* 3, 135-155

▪ Brennan, M. J. and Schwartz, E. S. (1982) "An Equilibrium Model of Bond Pricing and a Test of Market Efficiency." *Journal of Financial and Quantitative Analysis* 17, 301-329

▪ Courtadon, G. (1982). "The Pricing of Options on Default Free Bonds." *Journal of Financial and Quantitative Analysis* 17, 75-100

▪ Cox, J. C., Ingersoll Jr., J. E. and Ross, S. A. (1985a). "An Intertemporal General Equilibrium Model of Asset Prices." *Econometrica* 53, 363-385

▪ Cox, J. C., Ingersoll, J. E. and Ross S. A. (1985b). "A Theory of the Term Structure of Interest Rates." *Econometrica* 53, 385-407

▪ Dothan, L. (1978). "On the Term Structure of Interest Rates." *Journal of Financial Economics* 6, 59-69

▪ Fisher, M. and Zervos, D." Yield curve." Sta in Hal R. Varian. "Computational Economics and Finance. Modeling and Analysis with Mathematica." TELOS, The Electronic Library of Science, Santa Clara, CA, 1996

▪ Heath, D., Jarrow, R. and Morton, A. (1990). "Bond Pricing and the Term Structure of Interest Rates: A Discrete Time Approximation." *Journal of Financial and Quantitative Analysis* 25 (4), 419-440

▪ Heath, D., Jarrow, R. and Morton, A. (1991). "Contingent Claim Valuation with a Random Evolution of Interest Rates." *Review of Futures Markets*, 54-76

▪ Heath, D., Jarrow, R. and Morton, A. (January 1992). "Bond Pricing and the Term Structure of Interest Rates: A New Methodology for Contingent Claims Valuation." *Econometrica* 60, 77-105

▪ Ho, T. S. Y. and Lee, S. B. " Term Structure Movements and Pricing Interest Rate Contingent Claims." *The Journal of Finance* 41 (December 1986), 1011-1029

▪ Hull, J., and A. White (1990). "Pricing Interest Rate Derivative Securities." *Review of Financial Studies* 3(4), 573-592

▪ Langetieg, T. C. (1980). "A Multivariate Model of the Term Structure." *The Journal of Finance* 35(1), 71-97

▪ LIFFE Publications (1995). "Short Term Interest Rate Futures and Options." LIFFE Administration and Management, Cannon Bridge, London

▪ LIFFE Publications (1998). "Summary of Futures and Options Contracts." LIFFE Administration and Management, Cannon Bridge, London

▪ Longstaff, F. (1989). "A Nonlinear General Equilibrium Model of the Term Structure of Interest Rates." *Journal of Financial Economics* 23, 195-224

▪ Miller , R. M. "Option Valuation." Sta in Hal R. Varian. "Econometric Modeling with Mathematica." TELOS, The Electronic Library of Science, Santa Clara, CA, 1993

▪ Rendleman, R. and Bartter, B. (1980). "The Pricing of Options on Debt Securities." *Journal of Financial and Quantitative Analysis* 15, 11-24

▪ Richard, S. (1978). "An Arbitrage Model of the Term Structure of Interest Rates." *Journal of Financial Economics* 7, 38-58

▪ Shirakawa, H. (1991). "Interest Rate Option Pricing with Poisson-Gaussian Forward Rate Curve Processes." *Mathematical Finance* 1 (4), 77-94

▪ Turnbull, S. e Milne, F. (1991). "A Simple Approach to Interest Rate Option Pricing." *Review of Financial Studies* 4 (1), 87-120

▪ Vasicek, O. (1977). "An Equilibrium Characterization of the Term Structure." *Journal of Financial Economics* 5, 177-188